
Implementing Wake-Up on Key Stroke

FIGURE 1 - TWO KEY INTERFACE TO PIC16C5X

INTRODUCTION

In certain applications, the PIC16CXX is exercised only
when a key is pressed, eg. remote keyless entry.  In such
applications, the battery life can be extended by putting
the PIC16CXX to sleep during the inactive state and
when a key is pressed, the PIC16CXX wakes up, does
the task, and then goes back to sleep.

IMPLEMENTATION

The circuit in Figure 1 depicts an application with two
keys.  The PIC16C54 is normally in SLEEP mode
consuming very little operating current.  If either of the
two keys is pressed, the PIC16C5X ‘wakes up’, scans
the keys and turns on one of the two LED’s.  When SW1
is pressed, the green LED is turned on and when SW2
is pressed the red LED is turned on.  The LED’s are used
purely for demonstration purposes.  In real life applica-
tion, a transmission will be completed before putting the
PIC16C5X back in sleep.This example can be extended
to handle more than two keys.

In the sleep mode, the scan outputs (SCAN1 and SCAN2)
are both set to a low logic level.  In this state, the
capacitor C is fully charged and a high logic level is
present at the MCLR pin of the PIC16C5X.  When a key
is pressed, C discharges through either R2 or R3 (de-

pending on SW1 or SW2 being pressed) and the voltage
across C falls rapidly (approx. 1 ms), causing a low at the
MCLR pin of the PIC16C5X, which in turn causes the
PIC16C5X to wake up and enter its reset state.  In reset,
the SCAN1 and SCAN2 outputs default to a high imped-
ance mode, so the discharge path for capacitor C is
blocked and it charges to a high level through resistor
R1.  Note that the RC values have been chosen such,
that the discharge and charge cycles times are less than
the reset time for the PIC16C5X (approx. 18 ms), and
certainly far less than the minimum duration of a key-
press (approx. 50-100 ms).

After the reset cycle is completed, the code execution
momentarily takes the SCAN1 and SCAN2 outputs low
in order to sample the key stroke(s).  This does not cause
the capacitor to discharge since the duration of the low
is of the order of 10 micro seconds.

Once the keystroke function has been executed, the
program loops until the key has been released, sets the
SCAN1 and SCAN2 outputs low and “goes back to
sleep”.  Resistors R4-R8 are not required for functional-
ity, but are recommended to provide protection from
electrostatic discharge (ESD).  Switches SW1 and SW2,
when pressed may frequently pass ESD to the PIC16C54.

FIGURE 2 - TWO KEY SCAN/WAKE-UP TIMING DIAGRAM

FIGURE 3 - PIC16C5X INTERFACE TO 4 X 4 KEY MATRIX

MPASM 1.00 Released        WU.ASM   7-15-1994  13:10:54                 PAGE  1
Key Stroke Wake Up

LOC  OBJECT CODE     LINE SOURCE TEXT

                     0001                 TITLE        “Key Stroke Wake Up”
                     0002         LIST  P = 16C54,f=inhx8m
                     0003 ;******************************************************
                     0004 ;       Program demonstrating key stroke wake up for
                     0005 ;       the PIC16CXX. Program has been implemented for
                     0006 ;       two keys, but can be extended for more keys.
                     0007 ;       When SW1 is pressed a green LED lights up.
                     0008 ;       When SW2 is pressed a red LED lights up.
                     0009 ;******************************************************
                     0010 ;
                     0011 ; Define equates
                     0012 ;
0002                 0013         PC       EQU    2
0006                 0014         PORT_B   EQU    6
0002                 0015         SCAN1    EQU    2
0003                 0016         SCAN2    EQU    3
0000                 0017         SW1      EQU    0
0001                 0018         SW2      EQU    1
0004                 0019         GRN_LED  EQU    4
0005                 0020         RED_LED  EQU    5
0014                 0021         MSEC_20  EQU    D’20'
0008                 0022         DB1      EQU    8
0008                 0023         GP       EQU    8
0009                 0024         DB2      EQU    9
                     0025 ;
                     0026 ;PORT_B ASSIGNMENTS:
                     0027 ;       0 —> SW1       INPUT
                     0028 ;       1 —> SW2       INPUT
                     0029 ;       2 —> SCAN1     OUTPUT
                     0030 ;       3 —> SCAN2     OUTPUT
                     0031 ;       4 —> GRN_LED   OUTPUT
                     0032 ;       5 —> RED_LED   OUTPUT
                     0033 ;     6&7 —> ASSIGNED AS DUMMY OUTPUTS
                     0035 ;
                     0036 ;
                     0037         ORG      0
                     0038 ;
                     0039 START
0000 0910            0040         CALL    INIT_PORT_B     ;INITIALIZE PORT B
0001 0920            0041         CALL    DELAY           ;DELAY 20 MSECS
0002 0915            0042         CALL    SCAN_KEYS       ;GET KEY VALUES
0003 0028            0043         MOVWF   GP              ;SAVE IN RAM
0004 0608            0044         BTFSC   GP,SW1          ;SKIP IF SW1 NOT PRESSED
0005 0929            0045         CALL    TURN_GREEN_ON   ;ELSE DO ROUTINE
0006 0628            0046         BTFSC   GP,SW2          ;SKIP IF SW2 NOT PRESSED
0007 092B            0047         CALL    TURN_RED_ON     ;ELSE DO ROUTINE
                     0048 CHK_FOR_KEY
0008 0920            0049         CALL    DELAY           ;DELAY FOR 20 MSEC
0009 0915            0050         CALL    SCAN_KEYS       ;GET KEY HIT
000A 0F00            0051         XORLW   0               ;EXCL. OR WITH 0
000B 0743 0A08       0052         BNZ     CHK_FOR_KEY     ;KEY STILL PRESSED
                     0053                                 ;THEN LOOP
                     0054 NO_KEY_PRESSED
000D 0446            0055         BCF     PORT_B,SCAN1    ;SET SCAN LINES LOW
000E 0466            0056         BCF     PORT_B,SCAN2    ;       /
000F 0003            0057         SLEEP                   ;SLEEP
                     0058 ;
                     0060 ;
                     0061 INIT_PORT_B
0010 0C03            0062         MOVLW   B’00000011'     ; config RB0, 1 as i/p’s
0011 0006            0063         TRIS    PORT_B          ;  and RB2-7 as o/p’s
0012 0CFF            0064         MOVLW   0FFh
0013 0026            0065         MOVWF   PORT_B          ;DEFAULT VALUES FOR PORT_B
0014 0800            0066         RETLW   0               ;RETURN WITH NO ERROR

                     0067 ;
                     0068 ;This routine, scans two keys and returns the following:
                     0069 ;       0 if no key is pressed
                     0070 ;       1 if SW1 is pressd
                     0071 ;       2 if SW2 is pressed
                     0072 ;       3 if SW1 and SW2 are pressed
                     0073 ;
                     0074 SCAN_KEYS
0015 0446            0075         BCF     PORT_B,SCAN1    ;ENABLE SCAN FOR SW1
0016 0466            0076         BCF     PORT_B,SCAN2    ;EANBLE SCAN FOR SW2
0017 0C03            0077         MOVLW   B’00000011'     ;LOAD MASK IN W
0018 0146            0078         ANDWF   PORT_B,0        ;AND WITH PORT
0019 0546            0079         BSF     PORT_B,SCAN1    ;DISABLE SCAN
001A 0566            0080         BSF     PORT_B,SCAN2    ;       /
001B 01E2            0081         ADDWF   PC,1            ;GET OFFSET TO TABLE
001C 0803            0082         RETLW   3               ;SW1 AND SW2 PRESSED
001D 0802            0083         RETLW   2               ;SW2 PRESSED
001E 0801            0084         RETLW   1               ;SW1 PRESSED
001F 0800            0085         RETLW   0               ;NO KEY PRESSED
                     0086 ;
                     0087 ;DELAY, IS A APPROX. WAIT FOR 20.4mSECS, FOR A SYSTEM
                     0088 ;USING A 2 Mhz CRYSTAL CLOCK.
                     0089 DELAY
0020 0C14            0090         MOVLW   MSEC_20
0021 0028            0091         MOVWF   DB1
                     0092 DLY1
0022 0069            0093         CLRF    DB2
0023 02E8            0094         DECFSZ  DB1
0024 0A26            0095         GOTO    DLY2
0025 0800            0096         RETLW   0
                     0097 DLY2
0026 02E9            0098         DECFSZ  DB2             ;INNER LOOP = 1.02 MSEC.
0027 0A26            0099         GOTO    DLY2            ;       /
0028 0A22            0100         GOTO    DLY1
                     0101 ;
                     0102 ;
                     0103 TURN_GREEN_ON
0029 0486            0104         BCF     PORT_B,GRN_LED
002A 0800            0105         RETLW   0
                     0106 ;
                     0107 TURN_RED_ON
002B 04A6            0108         BCF     PORT_B,RED_LED
002C 0800            0109         RETLW   0
                     0110 ;
                     0111         END
                     0112
                     0113

MEMORY USAGE MAP (‘X’ = Used,  ‘-’ = Unused)

0000 : XXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXX XXXXXXXXXXXXX— ————————
0040 : ———————— ———————— ———————— ————————

All other memory blocks unused.

Errors   :    0
Warnings :    0

